

## Quatrième partie

### Utiliser la fonction conditionnelle SI

*Vous savez réaliser des formules et utiliser des fonctions... bien !! Dans certains tableaux, il y a des cellules dont le contenu n'est renseigné que sous certaines conditions (Imaginons un problème : calculer une prime dépendant d'une condition d'ancienneté). Pour résoudre les problèmes à choix multiples, on utilise la fonction SI.*

- Ouvrez le document **conditions** et activez la feuille **Fonctions Conditionnelles**

#### SITUATION 1

- Dans un tableau des notes, commencez par effectuer les différents calculs statistiques (Note la plus élevée, la plus basse et la moyenne des notes)

Il faut afficher les décisions concernant le résultat de l'examen, à savoir : les élèves dont la note est inférieure à 10 sont "refusés", les autres sont "admis".

**Syntaxe générale de la fonction SI :**

<b>=SI(CONDITION</b>	<b>;</b>	<b>ACTION 1</b>	<b>;</b>	<b>ACTION 2)</b>
	↓		↓	
Si la <b>CONDITION</b> est vraie	<b>ALORS</b>	<b>Exécuter</b> <b>ACTION 1</b>	<b>Sinon</b>	<b>Exécuter</b> <b>ACTION 2</b>

Pour formuler la condition, vous avez à votre disposition les opérateurs de comparaison.

#### Opérateurs de comparaison

Egal à	↔	=
Inférieur à	↔	<
Supérieur à	↔	>
Inférieur ou égal à	↔	<=
Supérieur ou égal à	↔	>=
Différent de	↔	<>

*Définissons tout d'abord les éléments :*

Condition :	La note est inférieure à 10
Action 1 :	L'élève est refusé
Action 2 :	L'élève est admis

Traduisons dans la syntaxe de la fonction :

<b>=SI(CONDITION</b>	<b>;</b>	<b>ACTION 1</b>	<b>;</b>	<b>ACTION 2)</b>
	↓		↓	
<b>=SI(B4&lt;10</b>	<b>;</b>	<b>"Refusé"</b>	<b>;</b>	<b>"Admis")</b>

Lorsque l'action consiste à afficher du **texte**, celui-ci doit être entre **guillemets**.

- Cliquez en **C4**
- Entrez la formule **=SI(B4<10;"Refusé";"Admis")**

	A	B	C	D	E
1	<b>TABLEAU DES NOTES</b>				
2					
3		Notes	Décision		Ve
4	BARTHOD	2,5	=si(b4		BARTH
5	DESHAYES	12,0	SI(test_logique; [valeur_si_vrai]; [valeur_si_faux])		

Remarquez qu'Excel affiche une info-bulle dont le contenu s'affiche en gras en fonction de l'état d'avancement de votre saisie comme sur l'image ci-dessous :

- Validez
- Recopiez cette formule vers le bas

Vérifiez que les résultats affichés correspondent au problème posé :

#### TABLEAU DES NOTES

	Notes	Décision
BARTHOD	2,5	Refusé
DESHAYES	12,0	Admis
DURRENBERGER	19,0	Admis
FREMOND	5,0	Refusé
GONET	18,0	Admis
GUILLOU	16,0	Admis
LEDO	8,0	Refusé
PORQUIER	13,0	Admis

Note la plus élevée	19,0
Note la plus basse	2,5
Moyenne des notes	11,7

**SITUATION 2**

Dans l'exemple précédent, on a vu que les actions 1 et 2 pouvaient consister à afficher du texte, vous allez maintenant apprendre à intégrer un calcul dans une fonction conditionnelle.

- Placez-vous sur le **TABLEAU DES PRIMES**

**TABLEAU DES PRIMES**

Vendeurs	Chiffre d'affaires	Primes
BARTHOD	25000	
DESHAYES	35000	
DURRENBERGER	41000	
FREMOND	21000	
GONET	38400	
GUILLOU	45000	
LEDO	19500	
PORQUIER	32000	

Vous devez calculer les primes selon le chiffre d'affaires :

A partir de 30 000 € de chiffres d'affaires, la prime représente 3% du CA, dans l'autre cas la prime sera de 1%.

*Définissons tout d'abord les éléments :*

Condition	: Le chiffre d'affaires est supérieur ou égal à 30 000
Action 1	: 3% du CA
Action 2	: 1% du CA

Traduisons dans la syntaxe de la fonction

<b>=SI(CONDITION</b>	<b>;</b>	<b>ACTION 1</b>	<b>;</b>	<b>ACTION 2)</b>
	<b>↓</b>		<b>↓</b>	
<b>=SI(G4&gt;=30000</b>	<b>;</b>	<b>G4*3%</b>	<b>;</b>	<b>G4*1%)</b>

- Cliquez en **H4**
- Entrez la formule **=SI(G4>=30000;G4\*3%;G4\*1%)**
- Validez
- Recopiez cette formule vers le bas
- Sauvegardez votre classeur

*Remarques :* Les formules placées dans les actions 1 et 2 ne sont pas entre guillemets puisqu'il s'agit de formules de calcul

**EXERCICE : Tableau comparatif**

- Placez-vous sur la feuille **Exercices** du classeur **conditions**
- Votre mission (si vous l'acceptez !) : faire apparaître en colonne C le chiffre le plus grand des colonnes A et B, en utilisant la fonction SI

N'oubliez pas de sauvegarder !!!

**SITUATION 3**

Dans les situations précédentes, vous avez vu que les actions 1 et 2 pouvaient consister à afficher du texte ou des formules, vous allez maintenant apprendre à gérer le fait qu'une cellule puisse être vide dans les actions et dans les conditions.

- Dans le classeur **Conditions**, sur la feuille **Fonctions Conditionnelles**, placez-vous sur le TABLEAU DES SALAIRES

**TABLEAU DES SALAIRES**

	Salaire	Taux de Prime	Prime
BARTHOD	853,71		
DESHAYES	1 143,37		
DURRENBERGER	1 454,36		
FREMOND	1 829,39		
GONET	1 494,00		
GUILLOU	1 189,10		
LEDO	1 524,49		
PORQUIER	990,92		

1° Vous devez afficher le taux de prime sachant que :  
A partir d'un salaire de 1 500 €, le taux de prime est de 2%

*Définissons tout d'abord les éléments :*

Condition :	Le salaire est supérieur ou égal à 1 500
Action 1 :	Afficher 2%
Action 2 :	Ne rien Afficher

Traduisons dans la syntaxe de la fonction

<b>=SI(CONDITION</b>	<b>;</b>	<b>ACTION 1</b>	<b>;</b>	<b>ACTION 2)</b>
	<b>↓</b>		<b>↓</b>	
<b>=SI(B22&gt;=1500</b>	<b>;</b>	<b>2%</b>	<b>;</b>	<b>"")</b>

**Attention :**

Ne rien faire dans une fonction conditionnelle se traduit toujours par ""

- Cliquez en **C22**
- Entrez la formule **=SI(B22>=1500;2%;"")**
- Validez
- Recopiez cette formule vers le bas

2° La prime est calculée si un taux de prime est affiché

$$\text{Prime} = \text{Salaire} \times \text{Taux de prime}$$

*Définissons tout d'abord les éléments :*

Condition:	Le taux de la prime est affiché
Action 1 :	Calculer la prime
Action 2 :	Ne rien Afficher

Traduisons dans la syntaxe de la fonction

<b>=SI(CONDITION</b>	<b>;</b>	<b>ACTION 1</b>	<b>;</b>	<b>ACTION 2)</b>
	<b>↓</b>		<b>↓</b>	
<b>=SI(C22&lt;&gt;"")</b>	<b>;</b>	<b>B22*C22</b>	<b>;</b>	<b>""</b> )

- Cliquez en **D22**
- Entrez la formule **=SI(C22<>"";B22\*C22;"")**
- Validez, et recopiez cette formule vers le bas

TABLEAU DES SALAIRES			
	Salaire	Taux de Prime	Prime
BARTHOD	853,71		
DESHAYES	1 143,37		
DURRENBERGER	1 454,36		
FREMOND	1 829,39	2%	36,59
GONET	1 494,00		
GUILLOU	1 189,10		
LEDÓ	1 524,49	2%	30,49
PORQUIER	990,92		

- Sauvegardez votre travail

**SITUATION 4**

Les conditions où les actions peuvent contenir des fonctions (MOYENNE, MIN, MAX, etc.)

- Dans le classeur **conditions**, sur la feuille **Fonctions Conditionnelles**, placez-vous sur le TABLEAU DES EXAMENS

<b>TABLEAU DES EXAMENS</b>		
	<b>Notes</b>	<b>Observations</b>
BARTHOD	2,5	
DESHAYES	12,0	
DURRENBERGER	19,0	
FREMOND	5,0	
GONET	18,0	
GUILLOU	16,0	
LEDO	8,0	
PORQUIER	13,0	

Lorsque la note dépasse la moyenne des notes, il faut afficher **++** dans la colonne observations, dans le cas contraire afficher **--**.

*Définissons tout d'abord les éléments :*

Condition :	La note de l'élève est supérieure à la moyenne
Action 1 :	Afficher ++
Action 2 :	Afficher --

Traduisons dans la syntaxe de la fonction

<b>=SI(CONDITION</b>	<b>;</b>	<b>ACTION 1</b>	<b>;</b>	<b>ACTION 2)</b>
	↓		↓	
<b>=SI(G22&gt;moyenne(\$G\$22:\$G\$29)</b>	<b>;</b>	<b>"++"</b>	<b>;</b>	<b>"--")"</b>

**Rappel :**

Les références de la fonction MOYENNE sont fixes, elles n'évoluent pas (la liste de notes va toujours de G22 à G29). C'est pourquoi la moyenne est en référence absolue.

- Cliquez en **H22**
- Entrer la formule  
=SI(G22>MOYENNE(\$G\$22:\$G\$29);"++";"--")
- Validez
- Recopiez cette formule vers le bas

**TABLEAU DES EXAMENS**

	<b>Notes</b>	<b>Observations</b>
BARTHOD	2,5	--
DESHAYES	12,0	++
DURRENBERGER	19,0	++
FREMOND	5,0	--
GONET	18,0	++
GUILLOU	16,0	++
LEDO	8,0	--
PORQUIER	13,0	++

- Sauvegardez votre classeur, imprimez puis fermez.

## Calculs courants

4<sup>ème</sup> partie**Synthèse n°8**

## Commissions des ventes

- Ouvrez le classeur **commis**

	Ventes Totales	Commission	Prime	Total dû
Gabriel Coste	5 488,16 €			
Tuyet Tran	9 909,19 €			
Pierre Cousson	5 206,13 €			
Nabil Fenjiro	7 851,12 €			
Yann Lasse	7 447,13 €			
Marie Toupé	7 532,51 €			
Odile Chirois	7 582,81 €			
Anne Claire	8 370,98 €			

- Déterminez les commissions et les primes
  - Commission : 10% des ventes totales
  - Prime : 150 € à partir de 7 000 € de ventes
- Sauvegardez et imprimez puis fermez ce classeur.

## Calculs courants

4<sup>ème</sup> partie**Synthèse n°9**

## Relevé de chèques

- Ouvrez le classeur **cheques**

## Gestion des chèques

Numéro de chèques	Montant	< 15	> = 15
N° 15487215	23,48 €		
N° 15487216	82,47 €		
N° 15487217	7,93 €		
N° 15487218	9,76 €		
N° 15487219	23,48 €		
N° 15487220	123,64 €		
N° 15487221	30,91 €		
TOTAL			
MOYENNE			
NOMBRE			

- Affichez le montant de chaque chèque dans la colonne correspondant à sa tranche, puis terminez les calculs au bas du tableau.
- Sauvegardez, imprimez puis fermez ce classeur.

**Synthèse n°10**

## Les élections

- Créez le tableau suivant :

**RESULTAT DES ELECTIONS**

NOM	VOIX	
BALLADUR	16145	
CHEMINADE	216	
CHIRAC	14200	
DEVILLIERS	2921	
HUE	8222	
JOSPIN	23855	
LAGUILLER	5538	
VOYNET	3666	

- Trouvez une formule permettant d'afficher "**ELU**" en face du candidat concerné.
- Sauvegardez puis refermez votre fichier.